pragma solidity >=0.5.12;


/**
 * @dev Interface of the ERC20 standard as defined in the EIP. Does not include
 * the optional functions; to access them see {ERC20Detailed}.
 */
interface IERC20 {
    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `recipient`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `sender` to `recipient` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address sender, address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);
}

/**
 * @dev Wrappers over Solidity's arithmetic operations with added overflow
 * checks.
 *
 * Arithmetic operations in Solidity wrap on overflow. This can easily result
 * in bugs, because programmers usually assume that an overflow raises an
 * error, which is the standard behavior in high level programming languages.
 * `SafeMath` restores this intuition by reverting the transaction when an
 * operation overflows.
 *
 * Using this library instead of the unchecked operations eliminates an entire
 * class of bugs, so it's recommended to use it always.
 */
library SafeMath {
    /**
     * @dev Returns the addition of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `+` operator.
     *
     * Requirements:
     * - Addition cannot overflow.
     */
    function add(uint256 a, uint256 b) internal pure returns (uint256) {
        uint256 c = a + b;
        require(c >= a, "SafeMath: addition overflow");

        return c;
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting on
     * overflow (when the result is negative).
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     * - Subtraction cannot overflow.
     */
    function sub(uint256 a, uint256 b) internal pure returns (uint256) {
        return sub(a, b, "SafeMath: subtraction overflow");
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting with custom message on
     * overflow (when the result is negative).
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     * - Subtraction cannot overflow.
     *
     * _Available since v2.4.0._
     */
    function sub(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        require(b <= a, errorMessage);
        uint256 c = a - b;

        return c;
    }

    /**
     * @dev Returns the multiplication of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `*` operator.
     *
     * Requirements:
     * - Multiplication cannot overflow.
     */
    function mul(uint256 a, uint256 b) internal pure returns (uint256) {
        // Gas optimization: this is cheaper than requiring 'a' not being zero, but the
        // benefit is lost if 'b' is also tested.
        // See: https://github.com/OpenZeppelin/openzeppelin-contracts/pull/522
        if (a == 0) {
            return 0;
        }

        uint256 c = a * b;
        require(c / a == b, "SafeMath: multiplication overflow");

        return c;
    }

    /**
     * @dev Returns the integer division of two unsigned integers. Reverts on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator. Note: this function uses a
     * `revert` opcode (which leaves remaining gas untouched) while Solidity
     * uses an invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     * - The divisor cannot be zero.
     */
    function div(uint256 a, uint256 b) internal pure returns (uint256) {
        return div(a, b, "SafeMath: division by zero");
    }

    /**
     * @dev Returns the integer division of two unsigned integers. Reverts with custom message on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator. Note: this function uses a
     * `revert` opcode (which leaves remaining gas untouched) while Solidity
     * uses an invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     * - The divisor cannot be zero.
     *
     * _Available since v2.4.0._
     */
    function div(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        // Solidity only automatically asserts when dividing by 0
        require(b > 0, errorMessage);
        uint256 c = a / b;
        // assert(a == b * c + a % b); // There is no case in which this doesn't hold

        return c;
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * Reverts when dividing by zero.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     * - The divisor cannot be zero.
     */
    function mod(uint256 a, uint256 b) internal pure returns (uint256) {
        return mod(a, b, "SafeMath: modulo by zero");
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * Reverts with custom message when dividing by zero.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     * - The divisor cannot be zero.
     *
     * _Available since v2.4.0._
     */
    function mod(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        require(b != 0, errorMessage);
        return a % b;
    }
}

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following 
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // According to EIP-1052, 0x0 is the value returned for not-yet created accounts
        // and 0xc5d2460186f7233c927e7db2dcc703c0e500b653ca82273b7bfad8045d85a470 is returned
        // for accounts without code, i.e. `keccak256('')`
        bytes32 codehash;
        bytes32 accountHash = 0xc5d2460186f7233c927e7db2dcc703c0e500b653ca82273b7bfad8045d85a470;
        // solhint-disable-next-line no-inline-assembly
        assembly { codehash := extcodehash(account) }
        return (codehash != accountHash && codehash != 0x0);
    }

    /**
     * @dev Converts an `address` into `address payable`. Note that this is
     * simply a type cast: the actual underlying value is not changed.
     *
     * _Available since v2.4.0._
     */
    function toPayable(address account) internal pure returns (address payable) {
        return address(uint160(account));
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     *
     * _Available since v2.4.0._
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        // solhint-disable-next-line avoid-call-value
        (bool success, ) = recipient.call.value(amount)("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }
}

/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for ERC20;` statement to your contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
 */
library SafeERC20 {
    using SafeMath for uint256;
    using Address for address;

    function safeTransfer(IERC20 token, address to, uint256 value) internal {
        callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to, value));
    }

    function safeTransferFrom(IERC20 token, address from, address to, uint256 value) internal {
        callOptionalReturn(token, abi.encodeWithSelector(token.transferFrom.selector, from, to, value));
    }

    function safeApprove(IERC20 token, address spender, uint256 value) internal {
        // safeApprove should only be called when setting an initial allowance,
        // or when resetting it to zero. To increase and decrease it, use
        // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
        // solhint-disable-next-line max-line-length
        require((value == 0) || (token.allowance(address(this), spender) == 0),
            "SafeERC20: approve from non-zero to non-zero allowance"
        );
        callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, value));
    }

    function safeIncreaseAllowance(IERC20 token, address spender, uint256 value) internal {
        uint256 newAllowance = token.allowance(address(this), spender).add(value);
        callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    function safeDecreaseAllowance(IERC20 token, address spender, uint256 value) internal {
        uint256 newAllowance = token.allowance(address(this), spender).sub(value, "SafeERC20: decreased allowance below zero");
        callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     */
    function callOptionalReturn(IERC20 token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves.

        // A Solidity high level call has three parts:
        //  1. The target address is checked to verify it contains contract code
        //  2. The call itself is made, and success asserted
        //  3. The return value is decoded, which in turn checks the size of the returned data.
        // solhint-disable-next-line max-line-length
        require(address(token).isContract(), "SafeERC20: call to non-contract");

        // solhint-disable-next-line avoid-low-level-calls
        (bool success, bytes memory returndata) = address(token).call(data);
        require(success, "SafeERC20: low-level call failed");

        if (returndata.length > 0) { // Return data is optional
            // solhint-disable-next-line max-line-length
            require(abi.decode(returndata, (bool)), "SafeERC20: ERC20 operation did not succeed");
        }
    }
}

contract LORVestingVault {
	using SafeMath for uint256;
	using SafeMath for uint16;
	using SafeERC20 for IERC20;

	uint256 internal constant SECONDS_PER_DAY = 86400;

	struct Grant {
		uint256 startTime;
		uint256 amount;
		uint16 vestingDuration;
		uint16 vestingCliff;
		uint16 daysClaimed;
		uint256 totalClaimed;
		address recipient;
	}

	IERC20 public token;
	uint256 private _totalSupply;
	uint256 private _totalGranted;
	uint256 private _totalClaimed;

	address public denMultiSig;
	uint256 public totalVestingCount;

	mapping(uint256 => Grant) public tokenGrants;
	mapping(address => uint256[]) private activeGrants;

	event TokensAdded(uint256 amount);
	event GrantAdded(address indexed recipient, uint256 vestingId);
	event GrantTokensClaimed(address indexed recipient, uint256 amountClaimed);
	event GrantRemoved(address recipient, uint256 amountVested, uint256 amountNotVested);
	event ChangedMultisig(address multisig);

	constructor(IERC20 _token) public {
		require(address(_token) != address(0));
		denMultiSig = msg.sender;
		token = _token;
	}

	modifier onlyDenMultiSig {
		require(msg.sender == denMultiSig, "not owner");
		_;
	}

	modifier onlyGrantRecipient(uint256 _grantId) {
		Grant storage tokenGrant = tokenGrants[_grantId];
		require(msg.sender == tokenGrant.recipient, "not grant recipient");
		_;
	}

	modifier onlyValidAddress(address _recipient) {
		require(
			_recipient != address(0) && _recipient != address(this) && _recipient != address(token),
			"not valid recipient"
		);
		_;
	}

	function totalSupply() public view returns (uint256) {
		return _totalSupply;
	}

	function totalGranted() public view returns (uint256) {
		return _totalGranted;
	}

	function totalClaimed() public view returns (uint256) {
		return _totalClaimed;
	}

	function deposit(uint256 amount) public onlyDenMultiSig {
		_totalSupply = _totalSupply.add(amount);
		token.safeTransferFrom(msg.sender, address(this), amount);
		emit TokensAdded(amount);
	}

	function addTokenGrant(
		address _recipient,
		uint256 _startTime,
		uint256 _amount,
		uint16 _vestingDurationInDays,
		uint16 _vestingCliffInDays
	) external onlyDenMultiSig {
		require(_totalGranted.add(_amount) <= _totalSupply, "add more tokens");
		require(_vestingCliffInDays >= 30, "less than 30 days");
		require(_vestingCliffInDays <= 365, "more than 1 year");
		require(_vestingDurationInDays <= 10 * 365, "more than 10 years");
		require(_vestingDurationInDays >= _vestingCliffInDays, "Duration < Cliff");

		uint256 amountVestedPerDay = _amount.div(_vestingDurationInDays);
		require(amountVestedPerDay > 0, "vested per day must be more than 0");

		Grant memory grant = Grant({
			startTime: _startTime == 0 ? currentTime() : _startTime,
			amount: _amount,
			vestingDuration: _vestingDurationInDays,
			vestingCliff: _vestingCliffInDays,
			daysClaimed: 0,
			totalClaimed: 0,
			recipient: _recipient
		});

		// add the amount to total granted tokens
		_totalGranted = _totalGranted.add(_amount);

		tokenGrants[totalVestingCount] = grant;
		activeGrants[_recipient].push(totalVestingCount);
		emit GrantAdded(_recipient, totalVestingCount);
		totalVestingCount++;
	}

	function getActiveGrants(address _recipient) public view returns (uint256[] memory) {
		return activeGrants[_recipient];
	}

	/// @notice Calculate the vested and unclaimed months and tokens available for `_grantId` to claim
	/// Due to rounding errors once grant duration is reached, returns the entire grant amount remaining
	/// Returns (0, 0) if cliff has not been reached
	function calculateGrantClaim(uint256 _grantId) public view returns (uint16, uint256) {
		Grant storage tokenGrant = tokenGrants[_grantId];

		// For grants created with a future start date, that hasn't been reached, return 0, 0
		if (currentTime() < tokenGrant.startTime) {
			return (0, 0);
		}

		// Check cliff was reached
		uint256 elapsedTime = currentTime().sub(tokenGrant.startTime);
		uint256 elapsedDays = elapsedTime.div(SECONDS_PER_DAY);

		if (elapsedDays < tokenGrant.vestingCliff) {
			return (uint16(elapsedDays), 0);
		}

		// If over vesting duration, all tokens vested
		if (elapsedDays >= tokenGrant.vestingDuration) {
			uint256 remainingGrant = tokenGrant.amount.sub(tokenGrant.totalClaimed);
			return (tokenGrant.vestingDuration, remainingGrant);
		} else {
			uint16 daysVested = uint16(elapsedDays.sub(tokenGrant.daysClaimed));
			uint256 amountVestedPerDay = tokenGrant.amount.div(uint256(tokenGrant.vestingDuration));
			uint256 amountVested = uint256(daysVested.mul(amountVestedPerDay));
			return (daysVested, amountVested);
		}
	}

	/// @notice Allows a grant recipient to claim their vested tokens. Errors if no tokens have vested
	/// It is advised recipients check they are entitled to claim via `calculateGrantClaim` before calling this
	function claimVestedTokens(uint256 _grantId) external onlyGrantRecipient(_grantId) {
		uint16 daysVested;
		uint256 amountVested;
		(daysVested, amountVested) = calculateGrantClaim(_grantId);
		require(amountVested > 0, "vested amount is 0");
		require(_totalSupply.sub(amountVested) >= 0, "not enough tokens");

		Grant storage tokenGrant = tokenGrants[_grantId];
		tokenGrant.daysClaimed = uint16(tokenGrant.daysClaimed.add(daysVested));
		tokenGrant.totalClaimed = uint256(tokenGrant.totalClaimed.add(amountVested));

		// add the amount to total claimed tokens
		_totalClaimed = _totalClaimed.add(amountVested);

		// remove the amount from the total supply
		_totalSupply = _totalSupply.sub(amountVested);

		token.safeTransfer(tokenGrant.recipient, amountVested);
		emit GrantTokensClaimed(tokenGrant.recipient, amountVested);
	}

	/// @notice Terminate token grant transferring all vested tokens to the `_grantId`
	/// and returning all non-vested tokens to the Den MultiSig
	/// Secured to the Den MultiSig only
	/// @param _grantId grantId of the token grant recipient
	function removeTokenGrant(uint256 _grantId) external onlyDenMultiSig {
		Grant storage tokenGrant = tokenGrants[_grantId];
		address recipient = tokenGrant.recipient;
		uint16 daysVested;
		uint256 amountVested;
		(daysVested, amountVested) = calculateGrantClaim(_grantId);

		require(_totalSupply.sub(amountVested) >= 0, "not enough tokens");
		uint256 amountNotVested = (tokenGrant.amount.sub(tokenGrant.totalClaimed)).sub(amountVested);

		tokenGrant.startTime = 0;
		tokenGrant.amount = 0;
		tokenGrant.vestingDuration = 0;
		tokenGrant.vestingCliff = 0;
		tokenGrant.daysClaimed = 0;
		tokenGrant.totalClaimed = 0;
		tokenGrant.recipient = address(0);

		// add the vested amount to total claimed tokens
		_totalClaimed = _totalClaimed.add(amountVested);

		// remove the vested amount from the total supply
		_totalSupply = _totalSupply.sub(amountVested);

		// remove the non-vested amount from the total granted
		_totalGranted = _totalGranted.sub(amountNotVested);

		token.safeTransfer(recipient, amountVested);
		emit GrantRemoved(recipient, amountVested, amountNotVested);
	}

	function currentTime() private view returns (uint256) {
		return block.timestamp;
	}

	function tokensVestedPerDay(uint256 _grantId) public view returns (uint256) {
		Grant storage tokenGrant = tokenGrants[_grantId];
		return tokenGrant.amount.div(uint256(tokenGrant.vestingDuration));
	}

	function changeMultiSig(address _denMultiSig) external onlyDenMultiSig onlyValidAddress(_denMultiSig) {
		denMultiSig = _denMultiSig;
		emit ChangedMultisig(_denMultiSig);
	}
}